
Referanslar
Dunit İle Unit Testine Giriş
DUnit yazılım testine Xtreme yaklaşımını desteklemek için dizayn edilmiş sınıfların bir framework’üdür.  Delphi 4 ve sonrasını destekler. 
Esas fikir;kodunuzu geliştirdiğiniz ya da değiştirdiğiniz zaman başka bir aşamada test etmek yerine,aynı anda gerekli testi de geliştirmenizdir. Testlerinizi güncel tutarak ve düzenli aralıklarla uygulayarak , kodunuzu daha güvenilir hale getirebilir,ve yapacağınız değişimlerin mevcut kodunuzu kırmadığından daha emin olabilirsiniz. Uygulamalarınız bu yolla kendilerini test ederler. 

DUnit bu testleri daha rahat organize etmeniz ve çalıştırmanız için sınıflar sağlar. DUnit testleri çalıştırmanız için iki seçenek sağlar :
· Bir GUI uygulaması olarak, testleri ve test uygulamarını ayrı ayrı kolayca seçmenizi sağlayarak,
· Bir konsol uygulaması olarak. 

DUnit esasında Kent Beck ve Erich Gamma tarafından Java için hazırlanmış JUnit framework’den esinlenerek yapılmıştır. Ama framework şu anda Delphi’ye özel daha güçlü bir araçtır. Orijinal Delphi portu  Juanco Añez tarafından oluşturulmuştur ve framework’ü SourceForge’daki  DUnit Group tarafından sürdürülmektedir. 

Arşiv İçerikleri
DUnit dağıtım arşivi kendi klasörüne yerleştirilmelidir çünkü dizin yapısı korunmuştur :
	Directory 
	Description 

	DUnit 
	  

	  
	src 
	Kütüphane kaynağı 

	  
	doc 
	Yardım dosyası,anasayfa ve MPL lisansı 

	  
	  
	images 
	Anasayfa image’leri. 

	
	
	API 
	Time2Help tarafından üretilmiş API dökümantasyonu 

	  
	Contrib 
	Eklenen modüller. 

	  
	  
	XPGen 
	Otomatik test durumları üretmek için bir araç 

	  
	tests 
	Framework’ün kendisi için test durumları. 

	  
	bin 
	Önceden derlenmiş kendi çalışabilen  GUI test çalıştırıcı. 

	  
	examples 
	  

	  
	  
	cmdline 
	DUnit’in komut satırından nasıl çağrılacağına dair örnekler. 

	  
	  
	collection 
	Java benzeri koleksiyon uygulamaları ve bunların DUnit test durumları. 

	  
	  
	registration 
	Test durumu registry sistemini kullanma. 

	  
	  
	structure 
	Testleri organize etmek için alternatif yollar. 

	  
	  
	  
	diffunit 
	Test durumlarını kendi unit’lerine yerleştirme. 

	  
	  
	  
	sameunit 
	Test durumlarını test edilen kodla aynı unit’e yerleştirme. 

	
	
	registry 
	Test durumlarıyla registry hizmeti oluşturmanın adım adım örneği. 

	  
	  
	embeddable 
	GUITestRunner’ın başka bir aracın formlarına nasıl iliştirileceğini gösterir.

	  
	  
	(...) 
	  

	  
	  
	TListTest 
	Delphi sınıfları için test durumları.TList nesnesi. 


SRC dizini aşağıdaki dosyaları içerir :
	File 
	Description 

	TestFramework.pas 
	Framework’ün kendisi. 

	TestExtensions.pas 
	Test durumlarını geliştirmek için kullanılabilecek dekoratör sınıflar. 

	GUITesting.pas 
	Kullanıcı arabirimini test etmek için sınıflar (Formlar ve Dialoglar). 

	TextTestRunner.pas 
	Testleri konsol modunda çalıştırmak için rutinler. 

	GUITestRunner.pas 
	Framework’e grafik kullanıcı arabirimi. 

	GUITestRunner.dfm 
	GUITestRunner Formu 


Framework dizini  framework unit’lerinin daha önceki versiyonlarının derlenmiş halini ve kütüphaneyi bir paketi bağlar gibi bağlamaya yarayan Delphi .DCP paket tanımlama dosyasını içerir.  (Uygun .bpl dosyası bin dizinindedir.). 

Başlangıç
DUnit kullanan programlar yazmak için, kaynak kodu yada önceden derlenmiş framework dizinleri unit arama yolunuza (path) eklenmelidir. Bunu Tools | Environment Options | Library  menüsünden , DUnit yollarını ekleyerek yapabilirsiniz. Örnek olarak : 

Alternatif olarak, DUnit yolunu project options’a ekleyebilirsiniz. 

İlk Test Projeniz 
Yeni bir uygulama oluşturun, ve unit1’i kapatın. Yeni projeyi (genelde test etmek istediğiniz proje ile aynı dizine kaydedilir.) Project1Test.dpr gibi bir isimle kaydedin.
File | New | Unit komutuyla yeni bir unit oluşturun(formsuz). Bu dosya test durumlarını içerecek dosyadır,bu yüzden Project1TestCases gibi bir isimle kaydedin. Interface uses kısmına  bir  TestFramework referansı ekleyin. 

TTestcase’den türetilmiş bir  TTestcasefirst sınfı deklare edin, ve bunu aşağıda gösterildiği gibi bir testfirst metodu ile uygulayın. Aşağıdaki initialization kısmına dikkat edin,bu kısım DUnit framework’ü kullanarak TTestcasefirst sınıfına başvurur. 

unit Project1TestCases;

interface

uses

 TestFrameWork;

type

 TTestCaseFirst = class(TTestCase)

 published

   procedure TestFirst;

 end;

implementation

procedure TTestCaseFirst.TestFirst;

begin

 Check(1 + 1 = 2, 'Catastrophic arithmetic failure!');

end;

initialization

 TestFramework.RegisterTest(TTestCaseFirst.Suite);

end.

Test edilecek sonuçlar check metodu çağrılarına geçilir. Burada 1+1’in 2 olduğunu  onaylıyoruz.  TestFramework.RegisterTest  prosedürü  framework’ün  test başvuru sistemindeki verilen teste başvurur.  

Şimdi projeyi çalıştırmadan önce,projenin kaynağını açmak için Project | View Source menüsünü seçin. TestFrameWork ve GUITestRunner’ı uses kısmına ekleyin. Normal uygulama kodunu kaldırın ve yerine aşağıdaki kodu ekleyin : 

program Project1Test;

uses

 Forms,

 TestFrameWork,

 GUITestRunner,

 Project1TestCases in 'Project1TestCases.pas';

{$R *.RES}

begin

 Application.Initialize;

 GUITestRunner.RunRegisteredTests;

end.

Şimdi programı çalıştırmayı deneyin. Eğer herşey yolunda giderse, bir ağaç yapısı altında ulaşılabilir testleri gösteren DUnit GUI’yi görmeniz gerekir. (şu anda sadece Testfirst). Run butonuna basmak testi çalıştırır.GUI checkbox’lara basarak test hiyerarşisinin kısımlarını enable ve disable etmenizi sağlayabilir ayrıca testleri ve dalları seçmek yada seçimi kaldırmak için ekstra butonlar içerir. 

Başka testler eklemek için   TTestCaseFirst sınıfına yeni metodlar ekleyin. TTestCase.Suite sınıf metodu test metodlarının aşağıdaki şartları sağlayıp sağlamadığını anlamak için RTTI (RunTime Type Information)’yi kullanır : 

· Test methodları parametre almazlar. 

· Test methodları published olarak deklare edilirler.. 

Not : DUnit bulduğu her metod için sınıfın ayrı bir örneğini oluşturur , böylece test metodları örnek veriyi paylaşamazlar.
İki test daha eklemek için, TestSecond ve TestThird, meotdları bu şekilde deklare edin : 

TTestCaseFirst = class(TTestCase)

published

  procedure TestFirst;

  procedure TestSecond;

  procedure TestThird;

 end;

...

procedure TTestCaseFirst.TestSecond;

begin

 Check(1 + 1 = 3, 'Deliberate failure');

end;

procedure TTestCaseFirst.TestThird;

var

 i : Integer;

begin

  i := 0;

  Check(1 div i = i, 'Deliberate exception');

end;
Programı tekrar çalıştırırsanız, testsecond’un başarısız olduğunu görürsünüz (yanında küçük mor bir kutu vardır), ve TestThird bir istisna çıkartır (yanındaki kutu kırmızıdır). Başarılı olan testlerin yanlarındaki kutu yeşil renkli olur. Çalıştırılmayan testler gri renktedir. Başarısız testlerin listesi panoda raporlanmıştır, ve herbirinin detaylarını aşağıdaki panoyu tıkladığınız zaman görebilirsiniz. 

Eğer programı IDE’den çalıştırıyorsanız, istisnaya geldiğinizde programın durduğunu görürsünüz. Muhtemelen bu durum DUnit’i kullanırken karşılaşmak istediğiniz bir durum değildir. İstisnalardaki bu kırılmayı Tools | Debugger Options | Language Exceptions diyaloğundan "Stop on Delphi Exceptions" seçeneğini kaldırarak engelleyebilirsiniz. 

Kurulum ve Kaldırma 

Testleri çalıştırmadan önce ve sonra yapılan birtakım ortak hazırlıklar ve toparlamalar vardır. Mesela bir sınıfı test ederken,öncelikle sınıfın bir örneğini oluşturursunuz , bazı denetlemeler yapar ve işiniz bittiği zaman  örneği serbest bırakırsınız(free). Eğer yapmanız gereken çok test varsa, her test metodunu sonlandırırken aynı kodları çalıştırırsınız. DUnit bu durumlar için TTestCase sınıfında  iki tane sanal metod tanımlar : SetUp ve TearDown.Bu metodlar her testten önce ve sonra çalıştırılırlar. Xtreme test jargonunda , bu durum  fixture olarak tarif edilir. 

Sıradaki örnek TtestCaseFirst’ü  TStringList  üzerinde birkaç test yapacak şekilde genişletiyor : 

interface

uses

 TestFrameWork,

 Classes;  // TstringList için gerekli
type

 TTestCaseFirst = class(TTestCase)

 private

   Fsl : TStringList;

 protected

   procedure SetUp; override;

   procedure TearDown; override;

 published

   procedure TestFirst;

   procedure TestSecond;

   procedure TestThird;

   procedure TestPopulateStringList;

   procedure TestSortStringList;

 end;

...

procedure TTestCaseFirst.SetUp;

begin

 Fsl := TStringList.Create;

end;

procedure TTestCaseFirst.TearDown;

begin

 Fsl.Free;

end;

procedure TTestCaseFirst.TestPopulateStringList;

var

 i : Integer;

begin

 Check(Fsl.Count = 0);

 for i := 1 to 50 do    // Iterate

   Fsl.Add('i');

 Check(Fsl.Count = 50);

end;

procedure TTestCaseFirst.TestSortStringList;

begin

 Check(Fsl.Sorted = False);

 Check(Fsl.Count = 0);

 Fsl.Add('You');

 Fsl.Add('Love');

 Fsl.Add('I');

 Fsl.Sorted := True;

 Check(Fsl[2] = 'You');

 Check(Fsl[1] = 'Love');

 Check(Fsl[0] = 'I');

end;

Test suites 

Büyük bir uygulamayı test ederken, TTestCase’den  türetilmiş birden fazla sınıf kullanmak isteyebilirisiniz. Bu yüksek seviyeli onaylamaları eklemek için, daha önceki örnekte yaptığımız gibi,onları basitçe initialization kısmına ekleyebilirsiniz. Başka zamanlarda, testlerinize daha fazla yapı eklemek isteyebilirsiniz. Bu sebepten, DUnit test suites’lerin oluşturulmasını destekler. Bu testler başka testleri de içerebilirler. Hatta başka test suites’leri de içerebilirler. (Bu , Composite dizayn modelinin bir uygulamasıdır.). 

TTestCaseFirst  sınıfında SetUp ve TearDown metodları aritmetik test metodlarında gereksiz yere çağrılmaktadırlar. Bu yüzden Tstringlist ile uğraşan diğer iki metodu kendi test durumlarına ayırmak daha iyi bir yaklaşım olacaktır. TTestCaseFirst’ü TTestArithmetic ve TTestStringlist şeklinde iki ayrı  sınıfa ayırarak bunu yapabiliriz: 

type

  TTestArithmetic = class(TTestCase)

  published

    procedure TestFirst;

    procedure TestSecond;

    procedure TestThird;

  end;

 TTestStringlist = class(TTestCase)

 private

   Fsl : TStringList;

 protected

   procedure SetUp; override;

   procedure TearDown; override;

 published

   procedure TestPopulateStringList;

   procedure TestSortStringList;

 end;

(Tabi ki implementations kısmını da güncellemelisiniz). 

Şimdi unit’in initialization kısmını aşağıdaki gibi değiştirin : 

RegisterTest('Simple suite', TTestArithmetic.Suite);

RegisterTest('Simple suite', TTestStringList.Suite);

Test Suite’leri adım adım oluşturmak
TestFramework unit’i test suite’leri uygulayan TTestSuite sınıfını içerir.Bu sayede test hiyerarşilerini daha net bir kodla oluşturabilirsiniz.
Sıradaki fonksiyon bir test suite oluşturur ve buna iki tane test sınıfı ekler : 

function UnitTests: ITestSuite;

var

 ATestSuite: TTestSuite;

begin

 ATestSuite := TTestSuite.create('Some trivial tests');

 ATestSuite.addTest(TTestArithmetic.Suite);

 ATestSuite.addTest(TTestStringlist.Suite);

 Result := ATestSuite;

end;

Bu fonksiyon şu şekilde de kodlanabilirdi : 

function UnitTests: ITestSuite;

begin

  Result := TTestSuite.Create('Some trivial tests',

                         [                 

                          TTestArithmetic.Suite,

                          TTestStringlist.Suite

                          ]);

end;

Yukarıdaki örnekte, TTestSuite yapıcısı parametre olarak geçilen testleri suite’ye ekler. 

Test Suite’i yukarıdakilerden hangisi ile oluşturursanız oluşturun başvurmak istediğiniz testlere aşağıdaki gibi erişebilirsiniz : 

initialization

   RegisterTest('Simple Test', UnitTests);

end.

GUITestRunner’ı çalıştırdığınız zaman yeni hiyerarşiyi görebilirsiniz. 

Diğer Özellikler
Testleri konsol modunda çalıştırma
Bazen test suite’lerimizi konsol modunda çalıştırmak faydalı olabilir, mesela bir Makefile içinden çağırmak gibi. Testleri konsol modunda çalıştırmak için, GUITestRunner yerine TextTestRunner’ı kullanan bir .dpr dosyası oluşturun ve {$APPTYPE CONSOLE} compiler direktifini ekleyin,ya da  IDE’de Project | Options | Linker | Generate console application seçeneğini  seçin. 

Yukarıdaki örneği bir konsol uygulaması olarak oluşturmak için, Project1TestConsole.dpr’i aşağıdaki şekilde oluşturun : 

{$APPTYPE CONSOLE}

program Project1TestConsole;

uses

 TestFrameWork,

 TextTestRunner,

 Project1TestCases in 'Project1TestCases.pas';

{$R *.RES}

begin

 TextTestRunner.RunRegisteredTests;

end.

Program çalıştığında şu şekilde bir çıktı oluşur : 

--

DUnit: Testing.

..F.E..

Time: 0.20

FAILURES!!!

Test Results:

Run: 5

Failures: 1

Errors: 1

There was 1 error:

1) TestThird: EDivByZero: Division by zero

There was 1 failure:

1) TestSecond

'..F.E..' stringine dikkat edin. Burada framework çalıştırdığı her test için bir nokta,her başarısız test için bir ‘F’ her istisna için bir ‘E’ yazmıştır. 

Bir başarısızlıkla karşılaşıldığında TextTestRunner’ın programı sıfırdan farklı bir çıkış koduyla sonlandırmasını rxbHaltOnFailures parametresini şu şekilde geçerek sağlayabilirsiniz :  
        TextTestRunner.RunRegisteredTests(rxbHaltOnFailures);

Test Suite’leri Makefile içinden çalıştırdığınız zaman programı sıfırdan farklı bir çıkış koduyla durdurmak çok kullanışlı olur. 

Uzantılar 

TextExtensions unit’i DUnit framework’ün işlevselliğini genişleten sınıflara sahiptir.Sınıfların çoğu, GoF (Gang of Four) "Patterns of Software Design(Yazılım Dizayn Modelleri)" ,kitabında tanımlandığı gibi dekoratör modelini kullanır. 

TRepeatedTest 

TRepeatedTest herhangi bir testi belli bir sayıda çalıştırmanızı sağlar.Mesela TTestArithmetic’deki TestFirst’ü 10 kere tekrar etmek için şu kodu yazabilirsiniz :  

uses

 TestFrameWork,

 TestExtensions, // TrepeatedTest için gerekli
 Classes;        // TStringList için gerekli
...

function UnitTests: ITest;

var

  ATestArithmetic : TTestArithmetic;

begin

 ATestArithmetic := TTestArithmetic.create('TestFirst');

 Result := TRepeatedTest.Create(ATestArithmetic, 10);

end;

TTestArithmetic yapılandırıcısına dikkat edin. 

ATestArithmetic := TTestArithmetic.create('TestFirst');

Burada tekrar edilecek test metodunun adını geçtim. Bu testin adını tam doğru olarak yazmaya dikkat edin.
Eğer bütün TTestArithmetic metodlarını tekrar ettirmek istiyorsanız onları bir suite’e yapıştırabilirsiniz : 
function UnitTests: ITest;

begin

 Result := TRepeatedTest.Create(ATestArithmetic.Suite, 10);

end;

Not : Bu kısımda Ttestsetup diye kısa bişey vardı.Herifler heralde anlaşılmasın diye yazmışlar nedir neye yarar hiçbişey anlamadım ondan yazmadım da.

Üçüncü Parti Uzantılar
XPGen

XPGen verilen bir sınıf için otomatik olarak test şablonu üreten bir araçtır.Bu araç Contrib\XPGen dizini altında bulunabilir. 

DUnitWizard

DUnitWizard  Paul Spain tarafından oluşturulmuş açık kaynaklı bir Delphi 5-7 plug-in’idir. Bu plug-in DUnit projeleri ve testleri için bir şablon oluşturur.Böylece siz yalnızca test yazmayla uğraşırsınız.Daha fazlası için : http://xpro.com.au/Freeware/DUnitWizard.htm 

Referanslar 

DUnit anasayfası  SourceForge’da (https://sourceforge.net/projects/dunit/) Son Kaynak, Mail listeleri, FAQ etc 

Delphi programları için Xtreme testing ( http://www.suigeneris.org/juanca/writings/1999-11-29.html) Juancarlo Añez's introduction to his DUnit classes, originally published on the Borland Community website. 

JUnit Test Infected: Programcılar test yazmayı severler  (http://www.junit.org/junit/doc/testinfected/testing.htm) JUnit’i anlatan iyi bir metin, DUnit’in temeli olan Java Framework’ü.
Basit Smalltalk Testing: Örneklerle (http://www.xprogramming.com/testfram.htm) Kent Beck'in orijinal sayfası. Hard going for non-Smalltalkers. 
